```
> restart:
  with(LinearAlgebra):
  d:=1;
```

$$d := 1 \quad (1)$$

We define the Lax matrix using Theorem 4.1

```
> checkLRed12:=-t[2*d+1]*lambda^(2*d):
  for k from 0 to d-1 do checkLRed12:=checkLRed12-t[2*d+1]*Qinfty
  [2*k]*lambda^(2*k): od:
  checkLRed12:=checkLRed12;
```

$$checkLRed12 := -t_3\,\lambda^2 - t_3\,Qinfty_0 \quad (2)$$

```
> checkLRed11:=t00/lambda+t00/Qinfty[0]*lambda^(2*d-1):
  for k from 1 to d do checkLRed11:=checkLRed11-Pinfty[2*d-2*k]*
  lambda^(2*k-1): od:
  for k from 1 to d-1 do for m from 0 to d-1-k do checkLRed11:=
  checkLRed11-Pinfty[2*m]*Qinfty[2*k+2*m]*lambda^(2*k-1): od: od:
  for k from 1 to d-1 do checkLRed11:=checkLRed11+t00/Qinfty[0]*
  Qinfty[2*k]*lambda^(2*k-1): od:
  checkLRed11:=checkLRed11;

  checkLRed22:=-checkLRed11:
```

$$checkLRed11 := \frac{t00}{\lambda} + \frac{t00\,\lambda}{Qinfty_0} - Pinfty_0\,\lambda \quad (3)$$

```
> NumcheckLRed21poleinfinity:=-checkLRed11^2/checkLRed12:
  for j from d to 2*d do for m from 0 to 2*d-j do
  NumcheckLRed21poleinfinity:=NumcheckLRed21poleinfinity+t[2*d+1-2*
  m]*t[2*j+2*m-2*d+1]*lambda^(2*j)/checkLRed12: od: od:
  checkLRed21poleinfinity:=0:
  for i from 1 to 100 do
  checkLRed21poleinfinity:=checkLRed21poleinfinity-(residue
  (NumcheckLRed21poleinfinity*lambda^(-i),lambda=infinity))*lambda^
  (i-1): od:

  checkLRed21:=checkLRed21poleinfinity;
```

$$checkLRed21 := \frac{\left(\frac{t00}{Qinfty_0} - Pinfty_0\right)^2}{t_3} - 2\,t_1 + t_3\,Qinfty_0 - t_3\,\lambda^2 \quad (4)$$

```
> checkLRedMatrix:=Matrix(2,2,0):
  checkLRedMatrix[1,1]:=checkLRed11:
  checkLRedMatrix[1,2]:=checkLRed12:
  checkLRedMatrix[2,1]:=checkLRed21:
```

```
checkLRedMatrix[2,2]:=checkLRed22:
checkLRedMatrix:
```

We define the general auxiliary matrix using Theorem 4.1

```
> checkARed12:=0:
for i from 0 to d-1 do checkARed12:=checkARed12-t[2*d+1]*nu[2*
i+1]*lambda^(2*d-2*i-1): od:
for k from 1 to d-1 do for i from 0 to k-1 do checkARed12:=
checkARed12-t[2*d+1]*nu[2*i+1]*Qinfty[2*k]*lambda^(2*k-2*i-1):
od: od:
checkARed12;

nu[2*d]:=simplify(-residue(lambda^(2*d-1)*checkARed12/checkL12,
lambda=infinity));
```

$$-t_3 \nu_1 \lambda$$

$$\nu_2 := 0 \tag{5}$$

```
> checkARed11:=0:
NumcheckARed11:=0:
for i from 0 to d-1 do NumcheckARed11:=NumcheckARed11+ nu[2*i+1]*
lambda^(-2*i-1)*checkLRed11: od:
for i from 1 to 100 do  checkARed11:=checkARed11-(residue
(NumcheckARed11*lambda^(-i),lambda=infinity))*lambda^(i-1): od:
checkARed11;

checkARed22:=-checkARed11:
```

$$\nu_1 \left(\frac{t00}{Qinfty_0} - Pinfty_0 \right) \tag{6}$$

```
> checkARed21:=(-t00)/t[2*d+1]/(Qinfty[0])^2*LQinfty[0]/lambda:
QuantitycheckARed21:=checkLRed11/checkLRed12*
(checkLRed11/checkLRed12*checkARed12-2*checkARed11):
for i from 1 to 100 do  checkARed21:=checkARed21-(residue
(QuantitycheckARed21*lambda^(-i),lambda=infinity))*lambda^(i-1):
od:
for i from 0 to 100 do  checkARed21:=checkARed21+(residue
(QuantitycheckARed21*lambda^(i),lambda=0))*lambda^(-i-1): od:
AdditionalTermRed:=t00*(1-t00)/t[2*d+1]/Qinfty[0]^2/lambda*nu[2*
d-1]:
for k from 1 to d-1 do AdditionalTermRed:=AdditionalTermRed+t00*(1
-t00)/t[2*d+1]/Qinfty[0]^2/lambda*nu[2*k-1]*Qinfty[2*k]: od:
checkARed21:=checkARed21+AdditionalTermRed:
NumcheckARed21:=0:
```

```
for k from d to 2*d-1 do for j from k+1 to 2*d do for m from 0 to
2*d-j do NumcheckARed21:=NumcheckARed21- t[2*d+1-2*m]*t[2*j+2*m-2*
d+1]*nu[2*j-2*k-1]*lambda^(2*k+1): od: od: od:
for i from 1 to 100 do  checkARed21:=checkARed21-(residue
(NumcheckARed21/(-checkLRed12)*lambda^(-i),lambda=infinity))*
lambda^(i-1): od:
checkARed21:=checkARed21:
series(checkARed21,lambda=infinity);
```

$$-t_3\,\nu_1\,\lambda+\frac{-\frac{t00\,LQinfty_0}{t_3\,Qinfty_0^2}+\frac{-2\,t00\,Pinfty_0\,Qinfty_0^3\,\nu_1+t00^2\,Qinfty_0^2\,\nu_1}{Qinfty_0^4\,t_3}+\frac{t00\,(1-t00)\,\nu_1}{t_3\,Qinfty_0^2}}{\lambda} \qquad (7)$$

```
> checkARedMatrix:=Matrix(2,2,0):
  checkARedMatrix[1,1]:=checkARed11:
  checkARedMatrix[1,2]:=checkARed12:
  checkARedMatrix[2,1]:=checkARed21:
  checkARedMatrix[2,2]:=checkARed22:
  checkARedMatrix:
```

We define the derivatives of the Lax matrices. The derivative relatively to times and Darboux coordinates is specific to d=1 and should be adapted for other values of d.

```
> dcheckARedMatrixdlambda:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do dcheckARedMatrixdlambda
  [i,j]:=diff(checkARedMatrix[i,j],lambda): od: od:
> LcheckLRedMatrix:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do
  LcheckLRedMatrix[i,j]:=diff(checkLRedMatrix[i,j],t[1])*alpha[1]+
  diff(checkLRedMatrix[i,j],Qinfty[0])*LQinfty[0]+diff
  (checkLRedMatrix[i,j],Pinfty[0])*LPinfty[0]: od:  od:
```

We define the compatibility equations and solve them. This is the main test for the formulas of Theorem 4.1. This part is again specific to d=1 and should be adapted for other values of d

```
> CompatibilityEquations:=simplify(
  dcheckARedMatrixdlambda-LcheckLRedMatrix+Multiply
  (checkARedMatrix,checkLRedMatrix)-Multiply
  (checkLRedMatrix,checkARedMatrix)   ):
> simplify(series(CompatibilityEquations[1,1],lambda=0)):
  series(simplify(series(CompatibilityEquations[1,1],lambda=
  infinity)),lambda=infinity):
  series(simplify(series(CompatibilityEquations[1,2],lambda=
  infinity)),lambda=infinity):
  series(simplify(series(CompatibilityEquations[2,1],lambda=
```

```
infinity)),lambda=infinity):
EQ1:=residue(CompatibilityEquations[1,1],lambda=0);
EQ2:=-residue(CompatibilityEquations[1,1]/lambda^2,lambda=
infinity);
EQ3:=-residue(CompatibilityEquations[1,1]/lambda^1,lambda=
infinity);
EQ4:=-residue(CompatibilityEquations[1,2]/lambda^2,lambda=
infinity);
EQ5:=-residue(CompatibilityEquations[1,2]/lambda^1,lambda=
infinity);
```

$$EQ1 := \frac{-2\, t00\, Pinfty_0\, Qinfty_0^2\, \nu_1 - t00\, LQinfty_0\, Qinfty_0 + t00\, Qinfty_0\, \nu_1}{Qinfty_0^2}$$

$$EQ2 := \frac{-2\, t_3^2\, \nu_1\, Qinfty_0^3 + \left(\left(-Pinfty_0^2 + 2\, t_3\, t_1\right)\nu_1 + LPinfty_0\right) Qinfty_0^2 - t00\, (-1 + t00)\, \nu_1}{Qinfty_0^2}$$

$$EQ3 := 0$$

$$EQ4 := 0$$

$$EQ5 := t_3 \left(2\, Pinfty_0\, Qinfty_0\, \nu_1 + LQinfty_0 - \nu_1\right) \tag{8}$$

```
> LQinfty[0]:=factor(solve(EQ1,LQinfty[0]));
  LPinfty[0]:=factor(solve(EQ2,LPinfty[0]));
  EQ1:=simplify(EQ1);
  EQ2:=simplify(EQ2);
  EQ3:=simplify(EQ3);
  EQ4:=simplify(EQ4);
  EQ5:=simplify(EQ5);
```

$$LQinfty_0 := -\nu_1 \left(2\, Pinfty_0\, Qinfty_0 - 1\right)$$

$$LPinfty_0 := \frac{\nu_1 \left(2\, t_3^2\, Qinfty_0^3 + Pinfty_0^2\, Qinfty_0^2 - 2\, t_1\, Qinfty_0^2\, t_3 + t00^2 - t00\right)}{Qinfty_0^2}$$

$$EQ1 := 0$$

$$EQ2 := 0$$

$$EQ3 := 0$$

$$EQ4 := 0$$

$$EQ5 := 0 \tag{9}$$

```
> series(simplify(series(CompatibilityEquations[2,1],lambda=
  infinity)),lambda=infinity);
```

$$-2\, t_3\, \nu_1 + 2\, \alpha_1 \tag{10}$$

In order to solve the entry (2,1) of the compatibility equations, we need to express the nu's in terms of the vector \alpha.

```
> solve(-2*nu[1]*t[3]+2*alpha[1],nu[1]);
```

(11)

$$\frac{\alpha_1}{t_3} \tag{11}$$

```
> MRed:=Matrix(d,d,0):
  for i from 1 to d do for j from 1 to i do
  MRed[i,j]:=t[2*d+1-2*i+2*j]:
  od: od:
  MRed;
  alphaVector:=Matrix(d,1,0):
  for i from 1 to d do alphaVector[i,1]:=alpha[2*d-2*i+1]/(2*d-2*
  i+1): od:
  alphaVector;
  nuVector:=Matrix(d,1,0):
  for i from 1 to d do nuVector[i,1]:=nu[2*i-1]: od:
  nuVector;
```

$$\begin{bmatrix} t_3 \end{bmatrix}$$
$$\begin{bmatrix} \alpha_1 \end{bmatrix}$$
$$\begin{bmatrix} \nu_1 \end{bmatrix} \tag{12}$$

```
> nu[1]:=alpha[1]/t[3];
  simplify(Multiply(MRed,nuVector)-alphaVector);
```

$$\nu_1 := \frac{\alpha_1}{t_3}$$
$$\begin{bmatrix} 0 \end{bmatrix} \tag{13}$$

```
> simplify(CompatibilityEquations);
```

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix} \tag{14}$$

This verifies that the relation between nu's and alpha's is correct. We check now that the formulas for the Hamiltonians are correct.

```
> for j from 0 to d-1 do
  Hinfty[2*j]:=simplify(-residue(lambda^(-2*j-1)*(checkLRed11^2+
  checkLRed12*checkLRed21+checkLRed12*diff(checkLRed11/checkLRed12,
  lambda)),lambda=infinity)): od:
  Hinfty[0]:=Hinfty[0];
```

$$Hinfty_0 := \frac{-t_3^2\, Qinfty_0^3 - Pinfty_0^2\, Qinfty_0^2 + 2\, t_1\, Qinfty_0^2\, t_3 + t00^2 + Pinfty_0\, Qinfty_0 - t00}{Qinfty_0} \tag{15}$$

```
> HamGeneralRed:=0:
  for k from 0 to d-1 do HamGeneralRed:=HamGeneralRed+nu[2*k+1]*
  Hinfty[2*k]: od:
```

```
HamGeneralRed:=simplify(HamGeneralRed);
```

$$HamGeneralRed := \frac{\left(-t_3^2\, Qinfty_0^3 - Pinfty_0^2\, Qinfty_0^2 + 2\, t_1\, Qinfty_0^2\, t_3 + t00^2 + Pinfty_0\, Qinfty_0 - t00\right)\alpha_1}{Qinfty_0\, t_3} \quad \textbf{(16)}$$

```
> simplify(LQinfty[0]-diff(HamGeneralRed,Pinfty[0]));
  simplify(LPinfty[0]+diff(HamGeneralRed,Qinfty[0]));
```

$$0$$
$$0 \quad \textbf{(17)}$$

```
> HamGeneralRedFunction:=unapply(HamGeneralRed,alpha[1]):
  HamRedt1:=simplify(HamGeneralRedFunction(1));
  VectorHamRed:=Matrix(d,1,0):
  VectorHamRed[1,1]:=1*HamRedt1:
  HVectorRed:=Matrix(d,1,0):
  for k from 1 to d do
  HVectorRed[k,1]:=Hinfty[2*d-2*k]: od:

  simplify(VectorHamRed-Multiply(MRed^(-1),HVectorRed));
```

$$HamRedt1 := \frac{-Qinfty_0^3\, t_3^2 - Pinfty_0^2\, Qinfty_0^2 + 2\, Qinfty_0^2\, t_1\, t_3 + t00^2 + Pinfty_0\, Qinfty_0 - t00}{Qinfty_0\, t_3}$$

$$\left[\begin{array}{c} 0 \end{array}\right] \quad \textbf{(18)}$$

```
> simplify(checkARed11);
  simplify(checkARed12);
  simplify(checkARed21);
```

$$\frac{\alpha_1\left(-Pinfty_0\, Qinfty_0 + t00\right)}{t_3\, Qinfty_0}$$
$$-\alpha_1\, \lambda$$
$$-\alpha_1\, \lambda \quad \textbf{(19)}$$

```
> series(simplify(checkLRed11),lambda);
  series(simplify(checkLRed12),lambda);
  expand(simplify(series(simplify(checkLRed21),lambda)));
```

$$t00\, \lambda^{-1} + \frac{-Pinfty_0\, Qinfty_0 + t00}{Qinfty_0}\, \lambda$$
$$-t_3\, Qinfty_0 - t_3\, \lambda^2$$
$$t_3\, Qinfty_0 + \frac{Pinfty_0^2}{t_3} - 2\, t_1 - \frac{2\, t00\, Pinfty_0}{t_3\, Qinfty_0} + \frac{t00^2}{t_3\, Qinfty_0^2} - t_3\, \lambda^2 \quad \textbf{(20)}$$

```
> expand(HamRedt1);
```

$$-t_3\,Qinfty_0^2 - \frac{Qinfty_0\,Pinfty_0^2}{t_3} + 2\,Qinfty_0\,t_1 + \frac{t00^2}{Qinfty_0\,t_3} + \frac{Pinfty_0}{t_3} - \frac{t00}{Qinfty_0\,t_3} \quad \textbf{(21)}$$

```
> LQinfty[0];
  expand(LPinfty[0]);
```

$$-\frac{\alpha_1\left(2\,Pinfty_0\,Qinfty_0 - 1\right)}{t_3}$$

$$2\,\alpha_1\,t_3\,Qinfty_0 + \frac{\alpha_1\,Pinfty_0^2}{t_3} - 2\,\alpha_1\,t_1 + \frac{\alpha_1\,t00^2}{t_3\,Qinfty_0^2} - \frac{\alpha_1\,t00}{t_3\,Qinfty_0^2} \quad \textbf{(22)}$$